In[6]:= **NotebookDirectory[]**

Out[6]= C:\Dropbox\Work\myweb\Courses\Math_pages\Math_430\

In[7]:= **NotebookFileName[]**

Out[7]= C:\Dropbox\Work\myweb\Courses\Math_pages\Math_430\Heat_dbc_v12.nb

The objective is to solve the heat equation

$$\frac{\partial u}{\partial t}(x, t) = \frac{\partial^2 u}{\partial x^2}(x, t) \text{ on } \{(x, t) \in \ \mid 0 \le x \le L,\ 0 \le t\},$$

subject to the conditions

$u(0, t) = 0,\ u(L, t) = 0,\ \ t \ge 0$ (these are the Dirichlet **BCs**)

$u(x, 0) = f(x),\ x \in [0, L]$ (this is **IC**)

As discussed in class the sequence of product solutions which satisfy the BCs is, with $n \in \mathbb{N}$,

$$\mathrm{Exp}\left[-\left(\frac{n\,\mathrm{Pi}}{L}\right)^{\square} t\right] \mathrm{Sin}\left[\frac{n\,\mathrm{Pi}}{L} x\right]$$

Then, to satisfy the IC, we form an infinite superposition of the above sequence of solutions with the Fourier coefficients which are integrals

$$\sum_{k=1}^{\infty}\left(\frac{2}{L}\int_0^L f(\xi)\,\mathrm{Sin}\left[\frac{k\,\mathrm{Pi}}{L}\xi\right] d\xi\right) \mathrm{Exp}\left[-\left(\frac{k\,\mathrm{Pi}}{L}\right)^2 t\right] \mathrm{Sin}\left[\frac{k\,\mathrm{Pi}}{L} x\right]$$

Now I implement this solution in *Mathematica* using symbolic integration. For simplicity, I assume that we always work on the interval $[0, \pi]$. I cannot work with infinite series, so the summation ends at some positive integer nn. In this way I obtain an approximation for the solution; larger nn, better approximation.

In[8]:=
```
Clear[uue];
uue[x_, t_, ff_, nn_] :=
 Module[{fcs},
  fcs = Table[2/Pi Integrate[ff[x] * Sin[k x], {x, 0, Pi}],
    {k, 1, nn}]; Sum[fcs[[k]] Exp[-k^2 t] Sin[k x], {k, 1, nn}]]
```

The above formula defines a function uue of four variables. The variables x and t are the space and the time from the partial differential equation, the variable ff is the function from the initial condition and nn is a positive integer stating how many terms in the partial sum of the series we want to use. Larger the positive integer nn better the accuracy, but slower the process of calculation. A tricky part here is the variable ff; it is a function and it has to be written as a pure function.

Pure function is a Mathematica construct in which a function is written with a “placeholder” variable # and the expression for the function ends with &. For example the square function is written as a pure function as

In[9]:=
```
(#^2) &
```

Out[9]=
```
#1^2 &
```

To use this function we enclose the independent variable in square brackets as we always do with functions in Mathematica.

In[10]:=
```
(#^2) &[3]
```
Out[10]= 9

In[11]:=
```
(#^2) &[a]
```
Out[11]= a^2

Pay attention how Pure Function is used in examples. For example for the constant function 1& (there is no variable # since it is a constant function) and with nn=10 we have

In[12]:=
```
uue[x, t, 1 &, 20]
```
Out[12]=

$$\frac{4\, e^{-t}\, \text{Sin}[x]}{\pi} + \frac{4\, e^{-9\,t}\, \text{Sin}[3\,x]}{3\,\pi} + \frac{4\, e^{-25\,t}\, \text{Sin}[5\,x]}{5\,\pi} + \frac{4\, e^{-49\,t}\, \text{Sin}[7\,x]}{7\,\pi} + \frac{4\, e^{-81\,t}\, \text{Sin}[9\,x]}{9\,\pi} + \frac{4\, e^{-121\,t}\, \text{Sin}[11\,x]}{11\,\pi} + \frac{4\, e^{-169\,t}\, \text{Sin}[13\,x]}{13\,\pi} + \frac{4\, e^{-225\,t}\, \text{Sin}[15\,x]}{15\,\pi} + \frac{4\, e^{-289\,t}\, \text{Sin}[17\,x]}{17\,\pi} + \frac{4\, e^{-361\,t}\, \text{Sin}[19\,x]}{19\,\pi}$$

We will explore this function later.

For a complicated function f(x) it might be slow or imposable to do symbolic integration, so I implemented the same formula using numeric integration.

In[13]:=
```
Clear[uun];
uun[x_, t_, ff_, nn_] :=
 Module[{fcsn},
  fcsn =
   Table[2/Pi NIntegrate[ff[x] * Sin[k x], {x, 0, Pi},
      WorkingPrecision → 20, AccuracyGoal → 8,
      PrecisionGoal → 12, MaxRecursion → 60], {k, 1, nn}];
  Chop[Sum[fcsn[[k]] * Exp[-k^2 t] Sin[k x], {k, 1, nn}]]
 ]
```

For example for the function $(\sin(x))^{1/5}$ and nn=10 we have

In[14]:=
```
uun[x, t, (Sin[#])^(1/5) &, 10]
```

Out[14]=
$$1.2014168421100297171\,e^{-t}\,\mathrm{Sin}[x] + 0.30035421052750742932\,e^{-9t}\,\mathrm{Sin}[3x] + 0.16172919028404246195\,e^{-25t}\,\mathrm{Sin}[5x] + 0.10781946018936164129\,6\,e^{-49t}\,\mathrm{Sin}[7x] + 0.079692644487789039223\,e^{-81t}\,\mathrm{Sin}[9x]$$

To explore the solutions we need to name each one of them with a special name, as follows (in the next example I name the solution uu0 and I keep the variables x and t since this is an approximation for the solution for a specific f(x). I do this naming with = sign since using the function uue with four variables would be very slow.

In[15]:=
```
uu0[x_, t_] = uue[x, t, 1 &, 20]
```

Out[15]=

$$\frac{4\,e^{-t}\,\text{Sin}[x]}{\pi}+\frac{4\,e^{-9\,t}\,\text{Sin}[3\,x]}{3\,\pi}+\frac{4\,e^{-25\,t}\,\text{Sin}[5\,x]}{5\,\pi}+\frac{4\,e^{-49\,t}\,\text{Sin}[7\,x]}{7\,\pi}+\frac{4\,e^{-81\,t}\,\text{Sin}[9\,x]}{9\,\pi}+\frac{4\,e^{-121\,t}\,\text{Sin}[11\,x]}{11\,\pi}+\frac{4\,e^{-169\,t}\,\text{Sin}[13\,x]}{13\,\pi}+\frac{4\,e^{-225\,t}\,\text{Sin}[15\,x]}{15\,\pi}+\frac{4\,e^{-289\,t}\,\text{Sin}[17\,x]}{17\,\pi}+\frac{4\,e^{-361\,t}\,\text{Sin}[19\,x]}{19\,\pi}$$

Now we explore visually how well the Fourier series approximates the function 1

In[16]:=
```
Plot[{1, uu0[x, 0]}, {x, 0, Pi}, PlotRange → {0, 1.3},
 PlotStyle → {{Thickness[0.01], Black},
   {Thickness[0.005], RGBColor[1, 0, 0]}}]
```

Out[16]=



In[17]:=
```
Plot3D[{uu0[x, t]}, {x, 0, Pi}, {t, 0., 2}, PlotPoints → 30,
  PlotRange → {0, 1.3}]
```

Out[17]=



In[18]:=
```
Manipulate[Plot[{N[uu0[x, t]]}, {x, 0, Pi}, PlotPoints → 30,
  PlotRange → {0, 1.3}], {{t, 0.0125}, 0, 3},
 ControlPlacement → Top]
```

Out[18]=



In[19]:=
```
(# (Pi - #)) &[2]
```

Out[19]= $2\ (-2 + \pi)$

Let us explore a function that satisfies the boundary conditions, like x(Pi-x)

In[20]:=
```
uu1[x_, t_] = uue[x, t, # (Pi - #) &, 20]
```

Out[20]=

$$\frac{8\,e^{-t}\,\mathrm{Sin}[x]}{\pi}+\frac{8\,e^{-9t}\,\mathrm{Sin}[3x]}{27\,\pi}+\frac{8\,e^{-25t}\,\mathrm{Sin}[5x]}{125\,\pi}+\frac{8\,e^{-49t}\,\mathrm{Sin}[7x]}{343\,\pi}+\frac{8\,e^{-81t}\,\mathrm{Sin}[9x]}{729\,\pi}+\frac{8\,e^{-121t}\,\mathrm{Sin}[11x]}{1331\,\pi}+\frac{8\,e^{-169t}\,\mathrm{Sin}[13x]}{2197\,\pi}+\frac{8\,e^{-225t}\,\mathrm{Sin}[15x]}{3375\,\pi}+\frac{8\,e^{-289t}\,\mathrm{Sin}[17x]}{4913\,\pi}+\frac{8\,e^{-361t}\,\mathrm{Sin}[19x]}{6859\,\pi}$$

Visually explore how good is the approximation (The given function is the black thicker graph, the approximation is the thinner yellow graph; the approximation is so visually so good that it is difficult to distinguish the graphs.)

In[21]:=
```
Plot[{x (Pi - x), uu1[x, 0]}, {x, 0, Pi}, PlotRange → {0, 3},
 PlotStyle → {{Thickness[0.01], Black},
   {Thickness[0.004], RGBColor[1, 1, 0]}}]
```

Out[21]=



However, when I plot the difference of the given function and its approximation we can see better what the error is.

In[22]:=
```
Plot[x (Pi - x) - uu1[x, 0], {x, 0, Pi}, PlotRange → Automatic,
 PlotStyle → {{Thickness[0.007], RGBColor[1, 0, 0]}}]
```

Out[22]=



In[23]:=
```
Plot3D[{uu1[x, t]}, {x, 0, Pi}, {t, 0, 5}, PlotRange → {0, 3}]
```

Out[23]=



Now explore a more complicated function which satisfies the boundary conditions:

In[24]:=
```
uu3[x_, t_] = uun[x, t, (Sin[#])^(1/5) &, 50]
```

Out[24]=
$$1.20141684211002971719\, e^{-t} \text{Sin}[x] + 0.30035421052750742932\, e^{-9t} \text{Sin}[3x] + 0.16172919028404246195\, e^{-25t} \text{Sin}[5x] + 0.10781946018936164129 6\, e^{-49t} \text{Sin}[7x] + 0.07969264448778903922 3\, e^{-81t} \text{Sin}[9x] + 0.06261564924040567368 5\, e^{-121t} \text{Sin}[11x] + 0.05123098574215009671 8\, e^{-169t} \text{Sin}[13x] + 0.04314188273023166033 0\, e^{-225t} \text{Sin}[15x] + 0.03712208513996677749 9\, e^{-289t} \text{Sin}[17x] + 0.03248182449747093031 7\, e^{-361t} \text{Sin}[19x] + 0.02880463681851195706 5\, e^{-441t} \text{Sin}[21x] + 0.02582484680280382358 4\, e^{-529t} \text{Sin}[23x] + 0.02336533758348917366 9\, e^{-625t} \text{Sin}[25x] + 0.02130369014965189379 3\, e^{-729t} \text{Sin}[27x] + 0.01955270191817365556 2\, e^{-841t} \text{Sin}[29x] + 0.01804864792446799000 7\, e^{-961t} \text{Sin}[31x] + 0.01674392638775946061 1\, e^{-1089t} \text{Sin}[33x] + 0.01560229504313949738 9\, e^{-1225t} \text{Sin}[35x] + 0.01459569536293694917 4\, e^{-1369t} \text{Sin}[37x] + 0.01370208136112448290 28\, e^{-1521t} \text{Sin}[39x] + 0.01290390186435994990 80\, e^{-1681t} \text{Sin}[41x] + 0.01218701842745106380 14\, e^{-1849t} \text{Sin}[43x] + 0.01153992010386959138 50\, e^{-2025t} \text{Sin}[45x] + 0.01095314450536774776 31\, e^{-2209t} \text{Sin}[47x] + 0.01041884477339858929 46\, e^{-2401t} \text{Sin}[49x]$$

In[25]:=
```
Plot[{(Sin[#])^(1/5) &[x], uu3[x, 0]}, {x, 0, Pi},
 PlotRange → {0, 1.2},
 PlotStyle → {{Thickness[0.01], Black},
   {Thickness[0.004], RGBColor[1, 0, 0]}}]
```

Out[25]=



In[26]:=
```
Plot3D[{uu3[x, t]}, {x, 0, Pi}, {t, 0, 2},
  PlotRange → {0, 1.2}]
```

Out[26]=